

Available Banks

The following subroutine will find all valid banks, determine the size of RamWorks memory, and save the results into BankTbl. BankTbl will have the number of 64K banks found, followed by the table of valid bank numbers. Keep in mind that the bank numbers are not necessarily linear. For example: 'BankTbl:08 00 01 02 03 04 05 06 07' indicates a 512K RamWorks containing banks 0 through 7.

```

BankSel EQU $C073
MaxMem EQU 3*16 Maximum desired, 3 megabytes in this example

*Write bank number to each bank
    STA $C009 ;Store in alternate zero page
    LDY #$7F ;Valid banks range $00 to $7F
FindBanks STY BankSel ;Go through each bank
    STY $00 ;Store the bank number
    TYA
    EOR #$FF
    STA $01 ;Second self-check
    DEY
    BPL FindBanks
*Read them back to find valid banks and save in table
    LDA #$00
    TAY
    TAX
FindThem STY BankSel ;Search through all banks
    STA BankSel+3
    CPY $00
    BNE NotOne ;Check bank number
    TYA
    EOR #$FF
    CMP $01 ;Check second double-check
    BNE NotOne
    INX
    TYA ;Found valid bank-save in table
    STA BankTbl,X
    CPX #MaxMem ;Found all banks to be used
    BCS Done
NotOne INY ;Go through all valid bank ranges
    BPL FindThem
*Ending routine
Done LDA #$00 ;Reset to video bank
    STA BankSel
    STA $C008
    STX BankTbl ;Size of Ram card
    LDA #$FF
    STA BankTbl+1,X ;Mark end of table
    JMP Continue

BankTbl DS MaxMem+2
Continue EQU *
```